Lab #9 Solution

CE 311K - McKinney

Code:

```
Option Explicit

Dim D As Single, e As Single, Re As Single

Private Sub Command1_Click()

    Dim xl As Single, xu As Single, xr As Single
    Dim ea As Single, tol As Single, test As Single
    Dim it As Integer, maxit As Integer, i As Integer
    i = 0

    CommonDialog1.ShowOpen
    Open CommonDialog1.FileName For Input As #1
    CommonDialog1.ShowSave
    Open CommonDialog1.FileName For Output As #2

    Print #2, "Calculation of friction factor by the bisection method"
    picOutput1.Print "Calculation of friction factor by the bisection method"

    Input #1, maxit, tol
    Print #2, "Max iterations = ", maxit, " Max error = ", tol
    picOutput1.Print "Maximum iterations = ", maxit, " Max error = ", tol

    For i = 1 To 2
        Input #1, D, e, Re
        Print #2, "Case ", i
        Print #2, "D = ", D; " e = ", e, " Re = ", Re
        picOutput1.Print "Case ", i
        picOutput1.Print "D = ", D; " e = ", e, " Re = ", Re
        xl = InputBox("Enter the lower bound on f for Case ", "Lower Bound")
        xu = InputBox("Enter the upper bound on f for Case ", "Upper Bound")
        Print #2, "XL = ", xl, " XU = ", xu
        picOutput1.Print "XL = ", xl, " XU = ", xu

        it = 0
        ea = 100
        While ((it <= maxit) And (ea > tol))
            it = it + 1
            xr = (xl + xu) / 2
            test = g(xl) * g(xr)
            If (test < 0) Then
                xu = xr
            Else
                xl = xr
            End If
            ea = Abs((xu - xl) / (xu + xl) * 100)
        Wend

        If (it >= maxit) Then
            Print #2, "Case Terminated!"
```

```
            Print #2, "No solution after ", it, " interations."
            Print #2, "Case ", i, " iterations exceeded maximum"
            picOutput1.Print "Case Terminated!"
            picOutput1.Print "No solution after ", it, " interations."
            picOutput1.Print "Case ", i, " iterations exceeded maximum"
        Else
            Print #2, "Case ", i, " solved successfully"
            Print #2, "f = ", xr, " .  Iterations = ", it
            Print #2, " Relative error = ", ea
            picOutput1.Print "Case ", i, " solved successfully"
            picOutput1.Print "f = ", xr, " .  Iterations = ", it
            picOutput1.Print "Relative error = ", ea
        End If
    Next
End Sub

Private Function g(X As Single) As Single
    g = 1 / Sqr(X) - 1.14 + 2# * (Log10(e / D + 9.35 / (Sqr(X) * Re)))
End Function

Private Function Log10(X As Single) As Single
    Log10 = Log(X) / Log(10)
End Function
```

Output:

```
Calculation of friction factor by the bisection method
Maximum iterations allowed =               1000
Relative error tolerance =                0.001

Case           1
D =            0.1  e =      0.0025        Re =          30000
XL =           0.001         XU =          1
Case           1             solved successfully
f =            5.411322E-02
Iterations =             20
Relative error =            8.811764E-04

Case           2
D =            0.1  e =      0.0001        Re =          5000000
XL =           0.00001       XU =          1
Case           2             solved successfully
f =            0.0196791
Iterations =             22
Relative error =            6.057696E-04
```